AI-AEC-13095 c. 1
NASA-CR-121204

THERMOELECTRIC PUMP PERFORMANCE ANALYSIS COMPUTER CODE

AEC Research and Development Report

Atomics International Division
Rockwell International

P.O. Box 309
Canoga Park, California 91304

NOTICE

Printed in the United States of America
Available from
National Technical Information Service
U.S. Department of Commerce
5285 Port Royal Road
Springfield, Virginia 22151
Price: Printed Copy $3.00; Microfiche $0.95

AI-AEC-13095
SNAP REACTOR
SNAP PROGRAM
M-3679-R69
C-926
NASA-CR-121204

THERMOELECTRIC PUMP PERFORMANCE ANALYSIS COMPUTER CODE

J. L. JOHNSON

Atomics International Division
Rockwell International
P.O. Box 309
Canoga Park, California 91304

CONTRACT: AT(04-3)-701
ISSUED: JUNE 11, 1973

FOREWORD

The work described here was done at the Atomics International Division of Rockwell International Corporation, under the direction of the Space Nuclear Systems Division, a joint AEC-NASA office. Project management was provided by NASA-Lewis Research Center and the AEC-SNAP Project Office.

DISTRIBUTION

This report has been distributed according to the category "Systems for Nuclear Auxiliary Power (SNAP) Reactor — SNAP Program," as given in the Standard Distribution for Classified Scientific and Technical Reports, M-3679, and according to the NASA Supplementary Report Distribution List at the end of this report.

CONTENTS

TABLE

FIGURES

ABSTRACT

This report presents the computer program that has been used to analyze and design dual-throat electromagnetic dc conduction pumps for the 5-kwe ZrH Reactor Thermoelectric System. In addition to a listing of the code and corresponding identification of symbols, a discussion of the bases for this analytical model is provided.

I. INTRODUCTION

The 5-kwe Reactor Thermoelectric System (5-kwe System) is a compact, reflector-controlled, nuclear heat source, utilizing thermoelectric conversion to produce 5 kwe of electrical power continuously for 5 years. The prototype design of this SNAP system was completed by Atomics International, a division of Rockwell International.

The 5-kwe System is illustrated in Figure 1; the principal components are a nuclear reactor, thermoelectric converter modules, dual-throat thermoelectric pump, heat rejection space radiator, volume accumulator units, interconnecting liquid-metal heat transfer primary and secondary coolant piping systems, and piping expansion joint units.

Thermal energy produced in the nuclear reactor is transferred by liquid metal, NaK (binary eutectic 22% sodium - 78% potassium alloy), circulated by the dual-throat thermoelectric pump through the reactor and thermoelectric modules of the primary loop, and through the thermoelectric modules and the radiator of the secondary loop.

Previous SNAP system activities have included the design, fabrication, and testing of various types of dc electromagnetic Faraday-type pumps and pump systems. To facilitate the design of these pumps, the equations describing performance have been collected and set into computer codes. The "standard" equations for predicting pump performance have been modified by experimental test results obtained from SNAP-type dc pumps.

This report presents a computer model, established specifically for determining minimum-weight dual-throat pump designs of the type shown in Figure 2.



72-N27-56-1A

Figure 1. 5-kwe Reactor Thermoelectric System



Figure 2. Prototype Thermoelectric Pump Assembly

II. PUMP COMPUTER PROGRAM APPLICATION

The computer program, printed in Section III, does not provide the optimum pump design, per se, but it furnishes the data required to make pump design point selections. DCPMP provides performance characteristics, and pump and pumping system weights, as a function of throat dimensions. Minimum pump weight and minimum pumping system weight do not occur at the same throat geometry, and neither geometry may provide the most desirable voltage, current, and efficiency characteristics. The pumping system weight includes the weight of the pump, the pump power supply (TE modules), associated support structure, and associated system components required to supply and reject heat. The weight of the thermoelectric power supply for pumps requiring from 1500 to 1800 amp was set at 24 lb, and a weight assessment of 8 lb/kw of heat rejection was made for the associated heat supply and rejection equipment. The pump weight includes a penalty of 40% to provide the supporting structure for the pump and the TE modules. Selected pump performance parameters, such as voltage, current, power input, pump weight, and pumping system weight, are computed for many throat sizes. Plotting these parameters as a function of throat dimensions provides the throat and pump size at minimum pumping system weight, as well as selected key performance characteristics at this configuration. Selection of a pump design point, using this technique, is discussed in Reference 1 for the prototype pump.

The computer program shown in this report makes a number of assumptions. These include: (1) identical current in each throat, (2) identical geometry in each throat, including transition pieces, (3) identical flow rate and pressure rise in each throat, and (4) identical flux density in each throat. In an actual pump design, the transitions may differ, the pressure - flow rate requirements may differ, and the flux density may differ. The requirements of each throat must be input separately, and the program run at the operating temperature of each throat, to determine which throat requires the greatest hydraulic pumping power, and therefore is dictating the pump size. In the prototype pump design of Reference 1, the hydraulic pumping requirements were higher in the primary throat; thus, the primary throat determined the pump size for the 5-kwe Reactor TE

System. If the pumping requirements are significantly different for the two throats, it may be desirable to utilize different sized throats and magnet assemblies; this would entail higher design and fabrication costs.

The expressions used in this code to account for effects such as throat current leakage, interference effects between the current flow and the magnetic fields, and armature reaction effects are not universal expressions and are very dependent upon throat-magnet-bus geometry. These expressions were determined from experimental data, and testing may be required to determine those parameters for pump designs that differ from that shown in Figure 2.

III. DCPMP AND PARAT COMPUTER CODES

Following are the computer listings for the pump analytical model, DCPMP, and the temperature-sensitive parameters, PARAT. Table 1 identifies the symbols used in these programs. See Reference 6 for basic computer terminology.

DCPMP

```
100 SCRATCH:PRIME:
110 SCRATCH:SECØND:
120 SCRATCH:TALLY:
130 READ H0,B0,C8,C9,D
140 READ D1,D2,D3,D4
150 READ T,T1,T2,L3,Y
160 DATA 645,11250,.02,.30,2.5
170 DATA .265,.284,.321,.290
180 DATA .028,.022,.25,32,.025
190 LET P1=1.09
200 INPUT:ØNE:D5,U,R1,R2,R3,Z
210 GØ TØ 230
220 INPUT:TWØ:D5,U,R1,R2,R3,Z
230 GØ TØ 240
240 FØR Q=4.83 TØ 4.83
250 FØR B=.5 TØ .7 STEP .1
260 FØR A=1.1 TØ 1.5 STEP .2
270 FØR X=2 TØ 2.6 STEP .3
280 IF A<1.5 THEN 300
290 GØ TØ 320
300 LET K=0
310 GØ TØ 410
320 IF A<=1.5 THEN 340
330 GØ TØ 360
340 LET K=1
350 GØ TØ 410
360 IF A<=2.25 THEN 380
370 GØ TØ 400
380 LET K=2
390 GØ TØ 410
400 LET K=3
410 LET I1=R1*(A-K*T1)/(B*X)
420 LET I2=R2*(A+2*T)/(2*T*X)
430 LET I3=R2*(2*T+K*T1)/(B*X)
440 LET I4=2.6*R1/B
450 LET I5=I2*I4/(I2+I4)
460 LET M1=5.714E-7*I5/(B*(I5+I1))
470 LET M2=3.687E-14/(D5*B↑2*(I5+I1))
480 LET M3=.96E-14/(D5*B*R1)
490 LET S=4*A*B/(3.1416*D↑2)
500 LET N=8.64E4*Q/(U*(A+B))
```

```
510 IF N>2.3E3 THEN 540
520 LET F1=1.1*16/N
530 GØ TØ 550
540 LET F1=1.1*7.9E-2*N↑-.25
550 LET C1=(C8+C9*(1-S)↑2)*1.3E-3/(D5*(A-K*T1)↑2*B↑2)
560 LET C2=2*F1*(B+(A-K*T1))*(X+.4)*1.3E-3/(D5*B↑3*(A-K*T1)↑3)
570 LET M4=C1+C2
580 LET I=2*((P1+M4*Q↑2)*(M2+M3)*Q)↑.5/M1
590 LET G=M1*I/(2*(M2+M3)*Q)
600 LET P1=M1*G*I-(M2+M3)*G↑2*Q-M4*Q↑2
610 LET E1=6.452E-8*G*Q/(D5*B)
620 LET V1=I5*(I*I1+E1)/(I1+I5)+I*I3
630 IF Z=2 THEN 660
640 PRINT:PRIME:A;B;X;I;G;Q;P1;V1;D5
650 GØ TØ 710
660 LET P2=P1
670 LET Q2=Q
680 LET V2=V1
690 LET D9=D5
700 PRINT:SECØND:Q2,P2,V2,D9
710 GØ TØ 720
720 NEXT X
730 NEXT A
740 NEXT B
750 NEXT Q
760 IF Z=2 THEN 780
770 GØ TØ 220
780 RESTØRE:PRIME:
790 RESTØRE :SECØND:
800 INPUT:PRIME:A,B,X,I,G,Q,P1,V1,D5
810 INPUT:SECØND:Q2,P2,V2,D9
820 LET V3=(V1+V2)/2
830 LET V4=V1+V2+V3
840 LET P7=I*V4
850 LET P8=.113*(P1*Q/D5+P2*Q2/D9)
860 LET Ø=P8*100/P7
870 LET L1=B+2*(T+Y)
880 LET F=5.1*L1+1.23
890 LET A1=F*G*1.05*A*X/BO
900 LET L2=1.3*L1*G*1.05/HO
910 LET W1=L2*A1*D1
920 LET W2=2*T2*A*X*D2
930 LET A3=R3*L3/(V3/I)
940 LET W3=L3*D3*A3
950 LET DO=2*A*B/(A+B)
960 LET L=(D-DO)/.21
970 LET W4=2*D4*T*(X*(A+B)+L*(A+B+1.571*D)) +D4*K*T1*B*X
980 LET C3=.031*(A-K*T1)*B*(X+.4)
990 LET C4=.031*2.0944*L*(D↑2/4+D*(A+B)/6.2832+(A+B)↑2/9.869)
1000 LET W5=C3+C4
1010 LET W6=D2*(L1+L2+2*T2+5*A)*A1*11.25/19
1020 LET W=1.4*(W3+2*(W1+W2+W4+W5+W6))
```

```
1030 LET W8=W+24
1040 LET W9=(8*P7)/34
1050 LET T8=W8+W9
1060 PRINT A;B;X;I;G;W;T8;0;V4
1070 PRINT:TALLY:A;B;X;W1;W2;W3;W4;W5;W6;W8;W9
1080 GØ TØ 800
1090 END
```

PARAT

```
100 SCRATCH:ØNE
110 SCRATCH:TWØ
120 PRINT "PRINT YØUR PRIMARY AND SECØNDARY THRØAT TEMP IN DEG F"
130 INPUT T1,T2
140 LET R1=(13.49+7.463E-3*T1+7.205E-6*T1↑2)*1.0E-6
150 LET R2=(27+.015*T1)*1.0E-6
160 LET R3=(.58+.00148*(T1+T2)/2)*1.0E-6
170 LET Z=1
180 LET D5=(54.27-.008349*T1)/1728
190 LET T=T1+460
200 LET U=10↑(.6663+(380.26/T)-.4158*(CLG(T)))
210 PRINT:ØNE:LNM(10000);D5;U;R1;R2;R3;Z;D
230 LET R1=(13.49+7.463E-3*T2+7.205E-6*T2↑2)*1.0E-6
240 LET R2=(27+.015*T2)*1.0E-6
250 LET D5=(54.27-.008349*T2)/1728
260 LET T=T2+460
265 LET Z=2
270 LET U=10↑(.6663+(380.26/T)-.4158*(CLG(T)))
280 PRINT:TWØ:LNM(10000);D5;U;R1;R2;R3;Z;D
290 PRINT " "
300 PRINT "DCPMP READY TØ RUN AT PRI AND SEC THRØAT TEMP ØF"
310 PRINT T1,"  &  ", T2
999 END
```

TABLE 1

CODE SYMBOLS AND IDENTIFICATION
(Sheet 1 of 3)

Code Symbol	Parameter	Units	Remarks*
HO	Maximum Field Intensity	Oe	Input
BO	Maximum Field Density	G	Input
C8	Throat Entrance Loss Coefficient	-	Input
C9	Throat Exit Loss Coefficient	-	Input
D	Piping Diameter	in.	Input
D1	Magnet Density	$lb/in.^3$	Input
D2	Pole Piece Density	$lb/in.^3$	Input
D3	Bus Density	$lb/in.^3$	Input
D4	Throat and Splitter Density	$lb/in.^3$	Input
T	Throat Wall Thickness	in.	Input
T1	Splitter Thickness	in.	Input
T2	Pole Piece Length	in.	Input
L3	Bus Length	in.	Input
Y	Magnet Air Gap	in.	Input
D5	NaK Density	$lb/in.^3$	PARAT
U	NaK Viscosity	lb/hr-ft	PARAT
R1	NaK Resistivity	Ω-in.	PARAT
R2	Throat Wall Resistivity	Ω-in.	PARAT
R3	Bus Resistivity	Ω-in.	PARAT
Z	Throat Number (1 or 2)	-	PARAT
P1	Net Pump Pressure Head (Primary)	psi	Input, PRIME
P2	Net Pump Pressure Head (Secondary)	psi	Input, PRIME
Q	NaK Flow Rate (Primary)	lb/sec	Input, PRIME
Q1	NaK Flow Rate (Secondary)	lb/sec	Input, PRIME
B	Throat Height	in.	Variable, PRIME
A	Throat Width	in.	Variable, PRIME
X	Throat Length	in.	Variable, PRIME
K	Number of Splitters	-	DCPMP
I1	Resistance Through NaK	Ω	DCPMP
I2	Resistance Through Side Walls	Ω	DCPMP
I3	Resistance Through End Walls and Splitters	Ω	DCPMP

TABLE 1

CODE SYMBOLS AND IDENTIFICATION
(Sheet 2 of 3)

Code Symbol	Parameter	Units	Remarks*
I4	End Loss Leakage Resistance Through NaK	Ω	DCPMP
I5	Total Bypass Leakage Resistance	Ω	DCPMP
M1	See Section IV	-	DCPMP
M2	See Section IV	-	DCPMP
M3	See Section IV	-	DCPMP
M4	See Section IV	-	DCPMP
S	Area Ratio, Throat to Pipe	-	DCPMP
N	Reynolds Number	-	DCPMP
F1	Friction Factor for Throat	-	DCPMP
C1	Entrance and Exit Hydraulic Losses	psi	DCPMP
C2	Throat Hydraulic Pressure Losses	psi	DCPMP
I	Total Current	amps	PRIME
G	Throat Flux Density	G	PRIME
E1	Back emf	v	DCPMP
V1	Voltage Across Primary Throat	v	PRIME
V2	Voltage Across Secondary Throat	v	SECOND
V3	Voltage Drop Across Bus	v	Output
V4	Total Voltage Drop Across Pump	v	Output
P7	Power Input	w	Output
P8	Power Output (Hydraulic)	w	Output
ϕ	Efficiency	%	Output
L1	Distance Between Pole Faces	in.	DCPMP
F	Magnet Leakage Factor	-	DCPMP
A1	Magnet Cross-Sectional Area	in.^{2}	DCPMP
L2	Magnet Length	in.	DCPMP
W1	Magnet Weight	lb	TALLY
W2	Pole Piece Weight	lb	TALLY
A3	Bus Cross-Sectional Area	in.	TALLY
W3	Bus Weight	lb	TALLY
DO	Equivalent Throat Diameter	in.	DCPMP

TABLE 1

CODE SYMBOLS AND IDENTIFICATION
(Sheet 3 of 3)

Code Symbol	Parameter	Units	Remarks*
L	Diffuser Length	in.	DCPMP
W4	Throat and Diffuser Weight	lb	TALLY
C3	NaK Weight in Throat at 70° F	lb	-
C4	NaK Weight in Diffusers at 70° F	lb	-
W5	NaK Weight at 70° F	lb	TALLY
W6	Yoke Weight	lb	TALLY
W	Pump and Structure Weight	lb	Output
W8	Total Pump + TE Converter Weight	lb	TALLY
W9	Associated System Weight at 8 lb/kw	lb	TALLY
T8	Total Pump System Weight	lb	Output

*Remarks

Input	Read in via data statements
PARAT	Calculated by PARAT Program, and input upon request from Data Files ONE and TWO
DCPMP	Calculated by DCPMP Program
PRIME	Primary Throat Characteristics Data File
SECOND	Secondary Throat Characteristics Data File
TALLY	Weights Data File
Output	Parameters selected for direct printout

IV. dc PUMP PERFORMANCE CHARACTERIZATION

Performance characteristics of a dc electromagnetic-type pump can be determined from the equivalent electrical schematic and the standard pressure-flow equations of Reference 2. For the SNAP-type dc pump, this equivalent electrical schematic was modified as shown in Figure 3, by including the resistance of the splitter, the throat walls, and the bus that are electrically in series with the current flowing through the NaK in the throat.



Figure 3. dc Pump Equivalent Electrical Schematic

The symbols used in Figure 3 and in the following discussion are identical to those used in the computer program DCPMP (Section III), to facilitate cross checking between it and this discussion. All symbols are defined and listed in Table 1, in the order in which they appear in the program. Also, selected equations are identified by line number (in parentheses), indicating their numerical location in the program. Therefore, from Figure 3, the voltage across the pump is:

$$V1 = \frac{I5}{I1 + I5} \times (I \times I1 + E1) + I \times I3 \qquad (620)$$

where:

$$I5 = \frac{I4 \times I2}{I4 + I2} \qquad (450)$$

From Reference 2, the back emf is:

$$E1 = 6.452 \times 10^{-8} \times G \times Q/(D5 \times B) \qquad (610)$$

As a result of the pump program pump design, development, and testing, an additional term, to account for eddy current losses, was added to the pressure-flow relationship obtained from Reference 2. This term was derived in Reference 3, and the value of the constant used with this expression was experimentally obtained. The complete pressure-flow relationship can now be expressed in the following form:

$$P1 = M1 \times G \times I - M2 \times G^2 \times Q - M3 \times G^2 \times Q - M4 \times Q^2 \qquad (600)$$

(Equation 1)

where:

$M1 \times G \times I$ = Gross pressure developed (psi)

$M2 \times G^2 \times Q$ = Pressure reduction due to back emf developed by fluid moving through a magnetic field (psi)

$M3 \times G^2 \times Q$ = Pressure reduction due to eddy current (end) losses (psi)

$M4 \times Q^2$ = Pressure reduction due to hydraulic losses in throat and in the entrance-exit transition sections (psi)

M1, M2, M3, and M4 are temperature-dependent constants, and are expressed as follows:

$$M1 = (5.714 \times 10^{-7}/B) \times \frac{I5}{I5 + I1} \qquad (460)$$

$$M2 = 3.687 \times 10^{-14}/[D5 \times B^2 \times (I5 + I1)] \qquad (470)$$

$$M3 = 0.96 \times 10^{-14}/(D5 \times B \times R1) \qquad (480)$$

$$M4 = C1 + C2 \qquad (570)$$

where:

$$C1 = [C8 + C9 \times (1 - S)^2] \times 1.3 \times 10^{-3}/[D5 \times (A - K \times T1)^2 \times B^2] \quad (550)$$

$$C2 = 2 \times F1 \times [B + (A - K \times T1)] \times (X + 0.4) \times 1.3 \times 10^{-3}/$$
$$[D5 \times B^3 \times (A - K \times T1)^3] \quad (560)$$

The maximum magnetic flux density in the throat is obtained by differentiating Equation 1 with respect to the flux, G. Thus

$$G = M1 \times I/[2 \times (M2 + M3) \times Q] \quad (590)$$

Substituting G back into Equation 1 and solving for I provides the current which will supply the required pressure and flow rate at this value of throat magnetic flux density:

$$I = 2 \times [(P1 + M4 \times Q^2) \times (M2 + M3) \times Q]^{1/2}/M1 \quad (580)$$

Higher flux densities (therefore longer magnets) will not provide greater pressure at this value of current and flow rate.

The remaining equations in the program, for calculating power, efficiency, sizes, and weights, are of the standard textbook variety, and can be obtained from Reference 4. The voltage drop across the bus is fixed at one half of the total voltage across the two throats. Efficiency calculations include both throats and the total length of the current bus ($\sim$32 in. long).

One of the key parameters in the analysis of the development pump (Reference 1), in relating the experimental results with predicted results, is the current passing through the NaK beyond the ends of the active throat. This is expressed as an end loss leakage resistance term in Line (440).

$$I4 = 2.6 \times R1/B$$

The constant used in this expression was obtained from Reference 3, but the geometry of the experimental pump for which this constant was obtained was not reported. The experimental data obtained from the development pump (Reference 1) provided a much higher value (lower current loss) for this term, because the magnet length exceeded the bus length by an amount (2.75 to 2.3 in.) that precluded much current from flowing outside of the magnetic field. Prediction of this term is one of the big uncertainties in the prediction of overall performance, and a careful examination of the magnet to bus relationship must be conducted by the pump designer.

During the design effort on the NASA PLPA program (Reference 5), another term was added to the DCPMP program to account for armature reaction caused by distortion in the magnetic field intensity between poles due to electrical current flow. For uncompensated pumps, this effect can seriously impair pump performance at high currents (see Reference 2). This term was used in the computer program in the design of the prototype pump in Reference 1 until experimental data obtained from the development pump showed it was not required for a pump of this configuration and current level. It is presently not in DCPMP; but, for higher current pumps, may be considered. This term (K) was used in the following manner in the pressure-flow relationship:

$$P1 = M1 \times G \times I \times K - (M2 + M3) \times G^2 \times Q - M4 \times Q^2$$

where:

$$K = 1 - \frac{2\pi \times I}{10 \times G \times B}\left[\coth B1 - \frac{1}{B1}\right]$$

$$B1 = \frac{2\pi \times Q \times X}{10^9 \times A \times B \times R1 \times D5}$$



Figure 4. Prototype dc Pump – Materials and Dimensions

V. PROGRAM USAGE AND OPERATION

If this and all associated programs have been erased from the computer files, then the following sequence must be performed (if the programs are in the user's file, start with Step 3):

1) Insert and save Programs DCPMP and PARAT, as listed in this report.
2) Create dummy programs called ONE, TWO, PRIME, SECOND, and TALLY. DCPMP and PARAT will not run unless this is done. Just provide name and save each.
3) Call PARAT and run. It will request primary and secondary throat temperatures. It computes NaK density and viscosity, and NaK, bus, and throat resistivities for each throat, storing the data in files ONE and TWO. The equations used in PARAT were supplied by NASA for the design effort conducted on the PLPA program (Reference 5).
4) Call DCPMP and check all input values, Lines 140 to 270, and then run. The values stored with this program are for the prototype pump discussed in Reference 1 and shown in Figures 2 and 4.

The code performs the following functions, in the order listed, for each input variable:

1) Determines the number of splitters, Lines 280-400
2) Computes resistances, Lines 410-450
3) Computes temperature-dependent constants, Lines 460-570
4) Computes current, Line 580
5) Computes magnetic flux densities, Line 590
6) Computes pressure (check of input pressure), Line 600
7) Computes voltage across throat, Line 620.

These equations are solved, and the results are stored in a data file, PRIME, for the primary throat, and then repeated for the secondary throat and stored in SECOND.

Additional equations are now solved in the following order for the primary throat current and magnetic flux density:

1) Bus voltage, Line 820
2) Total pump voltage, Line 830
3) Input electrical power, Line 840
4) Total hydraulic pumping power, Line 850
5) Pump efficiency, Line 860
6) Magnet dimensions, Lines 870-900
7) Weights, Lines 910-1050.

Selected parameters are printed out, Line 1060; for each input dimension and individual component weight data is stored in another data file, TALLY.

After the throat dimensions are selected, these and the corresponding value of current and flux density are fixed in the program, Lines 580 and 590. The flow rate, Line 240, is varied through the range of interest to determine the head-flow relationship for the primary throat. The flux density is reduced ~20% to obtain the head flow relationship for the secondary throat.

REFERENCES

1. J. L. Johnson, "Space Nuclear System Thermoelectric NaK Pump Development Summary Report," AI-AEC-13091 (to be published)

2. Liquid Metals Handbook, Na-NaK Supplement TID-5277 (July 1, 1955)

3. W. D. Leonard, "Thermoelectric Pump Design," NAA-SR-TDR-8841 (August 15, 1963)

4. T. T. Shimazaki, "Minimum Weight, Separate-Source Pump System," NAA-SR-TDR 10784 (December 7, 1964)

5. T. A. Moss et al., "Preliminary Design of a Primary Loop Pump Assembly (PLPA), Using Electromagnetic Pumps," AI-72-54 (September 1972)

6. "GE Series BASIC Language Reference Manual," CPB1470 (August 1968)

NASA Supplementary Report Distribution List for
AEC Contract No. AT(04-3)-701
LERC Order Nos. C21029 and C-21030

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Martin J. Saari (15)
M. S. 60-6

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Leonard W. Schopen (1)
M. S. 500-206

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Norman T. Musial (1)
M. S. 500-113

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Librarian (2)
M. S. 60-3

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: P. E. Foster (1)
M. S. 3-19

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention; G. Mervin Ault (1)
M. S. 3-5

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Robert E. English (1)
M. S. 500-201

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Henry O. Slone (1)
M. S. 501-6

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Donald Packe (4)
M. S. 500-201

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: B. T. Lundin (1)

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: B. Lubarsky (1)
M. S. 3-3

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Samuel J. Kaufman (3)
M. S. 49-2

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: V. F. Hlavin (1)
M. S. 3-10

NASA Lewis Research Center
21000 Brookpark Road
Cleveland, Ohio 44135
Attention: Neal T. Saunders (3)
M. S. 105-1

NASA Lewis Research Center
21000 Brookpark Road
Attention: Report Control Office
Cleveland, Ohio 44135
Attention: A. P. Dill (1)
M. S. 5-5

NASA Scientific and Technical
Information Facility
Acquisitions Branch
Post Office Box 33
College Park, Maryland 20740 (10)

National Aeronautics and
Space Administration
Washington, D. C. 20546
Attention: P. R. Miller (3)
Code RNP

National Aeronautics and
Space Administration
Washington, D. C. 20546
Attention: D. S. Gabriel (2)
Code NS 2

Jet Propulsion Laboratory
4800 Oak Grove Drive
Pasadena, California 91103
Attention: Library (1)

National Aeronautics and
Space Administration
Geo. C. Marshall Space Flight Center
Marshall Space Flight Center,
Alabama 35812
Attention: Library (1)

NASA Flight Research Center
P. O. Box 273
Edwards, California 93523
Attention: Library (1)

NASA Ames Research Center
Moffett Field, California 94035
Attention: Library (1)

NASA Goddard Space Flight Center
Greenbelt, Maryland 20771
Attention: Library (1)

NASA Langley Research Center
Langley Station
Hampton, Virginia 23365
Attention: Library (1)

NASA Manned Spacecraft Center
Houston, Texas 77058
Attention: Library (1)

Atomics International Division
Rockwell International

P.O. Box 309
Canoga Park, California 91304